


- Uda 1 GESTIRE FOGLI DI LAVORO
- Uda 2 **UTILIZZARE FUNZIONI**
- Uda 3 FOGLIO ELETTRONICO IN AZIENDA: LA FATTURAZIONE

OBIETTIVI UNITÀ

- Funzioni logiche: SE, E, O.
- Funzioni statistiche: CONTA.SE, MEDIA.SE, CONTA.PIÙ.SE.
- Funzioni matematiche: RAD.Q, POTENZA, SOMMA.SE, SOMMA.PIÙ.SE, MATRICE.SOMMA.PRODOTTO.
- Formule tridimensionali: SOMMA 3D o Consolida dati.
- Funzioni di ricerca e riferimento: CERCA.VERT, CERCA.ORIZZ.
- Visualizzare la sintassi delle formule: mostrare formule e funzioni inserite nel foglio di lavoro.



Il foglio elettronico dispone di numerose funzioni, divise per categorie, che permettono l'esecuzione di una vasta gamma di operazioni. Nel primo volume abbiamo preso in considerazione le funzioni più comuni tra quelle matematiche e statistiche, mentre qui ne prenderemo in esame altre tra quelle più usate in ambito scolastico.

AREA DIGITALE

- La formattazione condizionale

Inoltre, in un foglio di lavoro può essere utile **assegnare** regole di formattazione condizionale (**formattazione visiva** sotto forma di colori, icone o altro), che **facilitano** così l'analisi dei dati **risaltando la visualizzazione dei risultati numerici o determinati esiti di test condizionali**.

Le funzioni logiche

FUNZIONE	TIPO DI CALCOLO	SINTASSI
SE	Verifica una particolare condizione, restituendo un valore se la condizione è vera o un altro se la condizione è falsa. Se si vuole che al verificarsi della condizione falsa la cella rimanga vuota, nella sintassi della funzione, come ultimo argomento, si dovrà inserire "". I simboli delle virgolette (aperte e chiuse) restituiscono una stringa nulla e stanno a significare se falso non scrivere niente .	=SE(test;se_vero;se_falso), dove: - test corrisponde alla condizione da verificare (indirizzo di cella da esaminare); - se_vero indica l'argomento utilizzato se la condizione è vera (può essere un valore, un'etichetta o una formula); - se_falso indica l'argomento utilizzato se la condizione è falsa (può essere un valore, un'etichetta o una formula).
E	Permette di creare test condizionali composti che agiscono con gli operatori logici =, < e >. I test condizionali devono realizzarsi contemporaneamente .	=E(logico1;logico2;...), dove logico1 è l'argomento, o condizione, che viene verificato assieme a logico2 e a eventuali successivi.
O	Permette di creare test condizionali composti che agiscono con gli operatori logici =, < e >. I test condizionali si possono realizzare alternativamente .	=O(logico1;logico2;...), dove logico1 è l'argomento, o condizione, che viene verificato in alternativa a logico2 e a eventuali successivi.

Nel comporre i test condizionali, si utilizzano gli operatori logici descritti nella tabella sotto riportata.

OPERATORE LOGICO	SIGNIFICATO	ESEMPIO
=	Verifica se il contenuto della cella indicata è uguale al dato (valore o testo) espresso.	=SE(B10="Ammesso";.....)
<	Verifica se il contenuto della cella indicata è minore del valore espresso.	=SE(B10<5;.....)
>	Verifica se il contenuto della cella indicata è maggiore del valore espresso.	=SE(B10>5;.....)
<=	Verifica se il contenuto della cella indicata è minore o uguale al valore espresso.	=SE(B10<=5;.....)
>=	Verifica se il contenuto della cella indicata è maggiore o uguale al valore espresso.	=SE(B10>=5;.....)
<>	Verifica se il contenuto della cella indicata è diverso dal dato espresso. "<>" verifica se la cella contiene un dato qualsiasi (valore o testo), ovvero che non sia vuota.	=SE(B10<>"Ammesso";.....) =SE(B10<>0;.....)

ESERCIZIO

1 - Funzione logica SE semplice e annidata con E

Apri il file 2-Situazione pagamenti presente nei contenuti digitali del volume (hoepiscuola.it)

Un'azienda fa il punto sulla situazione riguardo ai saldi delle fatture emesse, in particolare dei pagamenti già scaduti o in prossimità di scadenza.

In D3 impostiamo la formula =B3+C3 per determinare la data di scadenza del pagamento di ciascuna fattura e copiamo la formula fino a D15 (Figura 1).

Nelle colonne E, F e G utilizziamo le funzioni logiche per impostare i test che andranno a verificare i risultati emersi.

D3	A	B	C	D	E
			=B3+C3		
1	Data verifica stato dei pagamenti		01/03/2020		
2	Ditta	Data fattura	Pagamento a	Data scadenza	Stato pagamenti
3	Rossi & C.	07/01/2020	60	07/03/2020	
4	Trimex	08/01/2020	60		
5	Rittals S.p.A.	10/01/2020	60		
6	Agrami S.p.A.	13/01/2020	60		
7	P.A.M. & A.	14/01/2020	60		
8	F.lli Alberti	16/01/2020	60		
9	Net & Net s.r.l.	20/01/2020	60		
10	Brown & Co.	22/01/2020	30		
11	Zeidwig S.n.c.	23/01/2020	60		
12	Harris & Clarke	23/01/2020	30		
13	F.lli Dai Molin	27/01/2020	60		
14	Interdis S.p.A.	29/01/2020	30		
15	ELDO s.r.l.	30/01/2020	60		

Figura 1

Inseriamo:

- in **E3** =SE(D3-\$C\$1<=0;"Inviare sollecito"; "") il test che confronti la data di ciascuna fattura con quella di verifica (01/03/2020). Se la data di scadenza del pagamento è stata oltrepassata (D3-\$D\$1<=0;), il test restituirà "Inviare sollecito", altrimenti lascerà la cella vuota ("). Il riferimento alla cella **C1** è inserito come assoluto per poter copiare la funzione fino a **riga 15**;
 - in **F3** =SE(E(D3-\$C\$1<=10;E3="");"Monitorare";"") la funzione che prevede due test (SE annidata con E): la data di scadenza del pagamento della fattura deve essere inferiore o uguale ai 10 giorni (D3-\$C\$1<=10;) e non deve essere previsto l'invio di sollecito, ossia la cella deve essere vuota (E3=""). Se entrambi i test risulteranno veri allora la funzione restituirà "Monitorare"; in caso contrario lascerà la cella vuota ("). Il riferimento alla cella **C1** è inserito come assoluto per poter copiare la funzione fino a **riga 15**;
 - in **G3** =SE(E(E3="";F3="");"Verificare dopo il 10/03";"") la funzione che prevede due test (SE annidata con E): non deve essere previsto l'invio di sollecito (E3="") e non deve essere monitorata (F3=""), ossia la celle devono essere vuote. Se entrambi i test risulteranno veri allora la funzione restituirà "Verificare dopo il 10/03"; in caso contrario lascerà la cella vuota (").
- Copiamo la formula fino a **G15** (Figura 2).

Figura 2

	A	B	C	D	E	F	G
1	Data verifica stato dei pagamenti		01/03/2020				
2	Ditta	Data fattura	Pagamento a (gg)	Data scadenza	Stato pagamenti		Note
3	Rossi & C.	07/01/2020	60	07/03/2020		Monitorare	
4	Trimex	08/01/2020	60	08/03/2020		Monitorare	
5	Rittals S.p.A.	10/01/2020	60	10/03/2020		Monitorare	
6	Agrami S.p.A.	13/01/2020	60	13/03/2020			Verificare dopo il 10/03
7	P.A.M. & A.	14/01/2020	60	14/03/2020			Verificare dopo il 10/03
8	F.lli Alberti	16/01/2020	60	16/03/2020			Verificare dopo il 10/03
9	Net & Net s.r.l.	20/01/2020	60	20/03/2020			Verificare dopo il 10/03
10	Brown & Co.	22/01/2020	30	21/02/2020	Inviare sollecito		
11	Zeldwig s.n.c.	23/01/2020	60	23/03/2020			Verificare dopo il 10/03
12	Harris & Clarke	23/01/2020	30	22/02/2020	Inviare sollecito		
13	F.lli Dal Molin	27/01/2020	60	27/03/2020			Verificare dopo il 10/03
14	Interdis S.p.A.	29/01/2020	30	28/02/2020	Inviare sollecito		
15	ELDO s.r.l.	30/01/2020	60	30/03/2020			Verificare dopo il 10/03

Elaboriamo il prospetto, utilizzando anche la **formattazione condizionale** nelle colonne E, F e G per evidenziare, con formattazioni diverse, le celle che contengono le etichette **Inviare sollecito**, **Monitorare** e **Verificare dopo il 10/03** (Figura 3).

Per eseguire tale operazione, in ogni colonna selezionare da **riga 3** a **riga 15**, quindi richiedere per ogni zona la formattazione condizionale impostando, per ogni criterio, una diversa evidenziazione, come mostrato nell'immagine.

Figura 3

	A	B	C	D	E	F	G
1	Data verifica stato dei pagamenti		01/03/2020				
2	Ditta	Data fattura	Pagamento a (gg)	Data scadenza	Stato pagamenti		Note
3	Rossi & C.	07/01/2020	60	07/03/2020		Monitorare	
4	Trimex	08/01/2020	60	08/03/2020		Monitorare	
5	Rittals S.p.A.	10/01/2020	60	10/03/2020		Monitorare	
6	Agrami S.p.A.	13/01/2020	60	13/03/2020			Verificare dopo il 10/03
7	P.A.M. & A.	14/01/2020	60	14/03/2020			Verificare dopo il 10/03
8	F.lli Alberti	16/01/2020	60	16/03/2020			Verificare dopo il 10/03
9	Net & Net s.r.l.	20/01/2020	60	20/03/2020			Verificare dopo il 10/03
10	Brown & Co.	22/01/2020	30	21/02/2020	Inviare sollecito		
11	Zeldwig s.n.c.	23/01/2020	60	23/03/2020			Verificare dopo il 10/03
12	Harris & Clarke	23/01/2020	30	22/02/2020	Inviare sollecito		
13	F.lli Dal Molin	27/01/2020	60	27/03/2020			Verificare dopo il 10/03
14	Interdis S.p.A.	29/01/2020	30	28/02/2020	Inviare sollecito		
15	ELDO s.r.l.	30/01/2020	60	30/03/2020			Verificare dopo il 10/03

ESERCIZIO

2 - Funzioni logiche SE e O annidate

Apri il file 2-Selezione personale presente nei contenuti digitali del volume (hoepitiscuola.it)

Una catena di hotel effettua una selezione di personale che abbia competenze in ambito dell'attività alberghiera da destinare all'accoglienza del cliente e alla raccolta delle prenotazioni. I criteri di selezione riguardano la conoscenza di almeno 4 lingue, oppure l'aver svolto mansioni di receptionist.

	A	B	C	D	E	F	G
1	Cognome	Nome	Settore	Lingue conosciute	Esito		
2	Viviani	Luca	Receptionist	6	Convocare		
3	Ghilardi	Luigi	Ristorazione	3			
4	Nalin	Giovanna	Amministrazione	5			
5	Zordi	Francesco	Receptionist	6			
6	Corlevic	Ivan	Servizio bar	3			
7	Maderna	Emiliano	Coordinamento sala	3			
8	Cantam	Alberto	Receptionist	6			
9	Rigotti	Giulia	Amministrazione	4			

Figura 4

Utilizziamo le funzioni logiche per impostare i test che andranno a verificare i requisiti richiesti.

- In **E2** inseriamo la funzione =SE(O(C2="Receptionist";D2>=4);"Convocare";"Non idoneo") che prevede due test (SE annidata con O): se il candidato esaminato ha svolto attività di receptionist (C2="Receptionist"); oppure conosce 4 o più lingue (D2>=4). Se uno dei due test risulterà vero allora la funzione restituirà "Convocare"; in caso contrario, restituirà "Non idoneo". Copiamo la funzione fino a **riga 9** (Figura 4).

	A	B	C	D	E
1	Cognome	Nome	Settore	Lingue conosciute	Esito
2	Viviani	Luca	Receptionist	6	Convocare
3	Ghilardi	Luigi	Ristorazione	3	Non idoneo
4	Nalin	Giovanna	Amministrazione	5	Convocare
5	Zordi	Francesco	Receptionist	6	Convocare
6	Corlevic	Ivan	Servizio bar	3	Non idoneo
7	Maderna	Emiliano	Coordinamento sala	3	Non idoneo
8	Cantam	Alberto	Receptionist	6	Convocare
9	Rigotti	Giulia	Amministrazione	4	Convocare

Figura 5

Elaboriamo il prospetto come mostrato, utilizzando anche la **formattazione condizionale** nelle zone E2:E9 per evidenziare le celle che contengono l'etichetta **Convocare** (Figura 5).

Le funzioni statistiche

FUNZIONE	TIPO DI CALCOLO	SINTASSI
CONTA.SE	In un intervallo, conta il numero di celle che soddisfano un singolo criterio specificato .	=CONTA.SE(intervallo;criteri), dove: • intervallo è la zona di celle da confrontare per il conteggio; • criteri rappresenta il test che esplicita quali celle verranno contate.
MEDIA.SE	Calcola il valore medio tra quelli presenti in un intervallo di celle che soddisfano un singolo criterio specificato .	=MEDIA.SE(intervallo;criterio;[int_m]), dove: • intervallo è la zona di celle da confrontare in base al criterio; • criterio rappresenta il test che esplicita i contenuti che le celle devono soddisfare perché siano considerate. Può essere scritto in forma di numero, espressione o testo; • int_m (opzionale) è l'intervallo di celle da utilizzare per il calcolo della media se le corrispondenti confrontate (intervallo) soddisfano i criteri di selezione.
CONTA.PIÙ.SE	Conta le celle di un intervallo che soddisfano tutti i criteri impostati .	=CONTA.PIÙ.SE(intervallo1;criteri1; intervallo2;criteri2;...), dove: • intervallo1 (obbligatorio) è il primo intervallo in cui eseguire il conteggio valutando i criteri associati; • criteri1 (obbligatorio) sono criteri in forma di numero, espressione, riferimento di cella o testo che determinano quali celle nell'argomento intervallo1 saranno conteggiate; • intervallo2;criteri2;... sono intervalli aggiuntivi e criteri associati.

ESERCIZIO

3 - Funzioni statistiche

Apri il file 2-Statistiche mezzi trasporto presente nei contenuti digitali del volume (hoepitiscuola.it)

Nell'ambito di un'indagine sui mezzi di trasporto usati dagli studenti universitari veronesi viene chiesto a un campione di 30 studenti, di età compresa tra i 19 e i 27 anni, quale mezzo di trasporto usano per recarsi in università. Dai risultati emersi ricaviamo alcuni dati statistici. Inseriamo:

- in H2 =CONTA.SE(\$D\$2:\$D\$31;"a piedi") per contare quanti studenti vanno a piedi;
- in G3 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"a piedi";\$C\$2:\$C\$31;"F") per contare quante sono le femmine che vanno a piedi;
- in G4 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"a piedi";\$C\$2:\$C\$31;"M") per contare quanti sono i maschi che vanno a piedi;
- in H6 =CONTA.SE(\$D\$2:\$D\$31;"bicicletta") per contare quanti studenti vanno in bicicletta;
- in G7 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"bicicletta";\$C\$2:\$C\$31;"F") per contare quante femmine usano la bicicletta;
- in G8 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"bicicletta";\$C\$2:\$C\$31;"M") per contare quanti maschi usano la bicicletta;
- in H10 =CONTA.SE(\$D\$2:\$D\$31;"moto") per contare quanti studenti usano la moto;
- in G11 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"moto";\$C\$2:\$C\$31;"F") per contare quante femmine usano la moto;
- in G12 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"moto";\$C\$2:\$C\$31;"M") per contare quanti maschi usano la moto;
- in H14 =CONTA.SE(\$D\$2:\$D\$31;"auto") per contare quanti studenti usano l'auto;
- in G15 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"auto";\$C\$2:\$C\$31;"F") per contare quante femmine usano l'auto;
- in G16 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"auto";\$C\$2:\$C\$31;"M") per contare quanti maschi usano l'auto;
- in H18 =CONTA.SE(\$D\$2:\$D\$31;"mezzo pubblico") per contare quanti studenti usano il mezzo pubblico;
- in G19 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"mezzo pubblico";\$C\$2:\$C\$31;"F") per contare quante femmine usano il mezzo pubblico;
- in G20 =CONTA.PIÙ.SE(\$D\$2:\$D\$31;"mezzo pubblico";\$C\$2:\$C\$31;"M") per contare quanti maschi usano il mezzo pubblico;
- in I22 =MEDIA.SE(D2:D31;"mezzo pubblico";B2:B31) per calcolare l'età media degli studenti che usano il mezzo pubblico;
- in I24 =MEDIA(B2:B31) per ottenere l'età media degli studenti intervistati.

Elaboriamo la tabella come mostrato (Figura 6).

Figura 6

G20		=CONTA.PIÙ.SE(\$D\$2:\$D\$31;"mezzo pubblico";\$C\$2:\$C\$31;"M")	
	A	B	C
1	Nome	Età	Sesso
2	Marco	19	M
3	Luca	21	M
4	Alberto	27	M
5	Giulia	26	F
6	Annachiara	23	F
7	Francesco	25	M
8	Giovanna	24	F
9	Henry	19	M
10	Sara	21	F
11	Matilde	19	F
12	Carlotta	21	F
13	Federico	26	M
14	Jennifer	22	F
15	Roberto	21	M
16	Karl	24	M
17	Francesca	21	F
18	Alice	23	F
19	Gaetano	25	M
20	Jonathan	21	M
21	Cinzia	25	F
22	Akahito	24	M
23	Anna	24	F
24	Angelo	27	M
25	Lorenzo	24	M
26	Michele	27	M
27	Stefano	24	M
28	Lucia	19	F
29	Loredana	20	F
30	Ygor	25	M
31	Denis	23	M

CURIOSITÀ

Come puoi notare, nelle funzioni inserite in H2 e in G3 sono stati applicati i riferimenti assoluti. Questo per agevolare la copia di tali funzioni nelle celle sottostanti che considerano sempre le stesse zone (D2:D31 e C2:C31); aggiornando poi i relativi criteri, si eviterà così di doverle riscrivere ogni volta.

Le funzioni matematiche

FUNZIONE	TIPO DI CALCOLO	SINTASSI
RADQ	Calcola la radice quadrata di un numero.	=RADQ(numero)
POTENZA	Restituisce il risultato di un numero elevato a potenza.	=POTENZA(num;potenza), dove: • potenza può essere 2 (quadrato), 3 (cubo) ecc.
MATR.SOMMA.PRODOTTO	Opera su due o più gruppi di valori restituendo la somma dei relativi prodotti.	=MATR.SOMMA.PRODOTTO(matrice1;matrice2;...), dove: • matrice1, matrice2, ... sono le zone contenenti valori da moltiplicare tra loro e poi da sommare.
SOMMA.SE	Verifica ogni cella utilizzando uno specifico test condizionale e permette di includere nella somma quelle celle dell'intervallo indicato che soddisfano la condizione riportata nei criteri previsti.	=SOMMA.SE(intervallo;criterio;int_somma), dove: • intervallo corrisponde all'intervallo di celle da verificare e la funzione controlla quali soddisfano la condizione espressa nel criterio; • criterio specifica il test condizionale. È dato dai criteri che determinano le corrispondenze delle celle da sommare; • int_somma indica le celle da sommare, se soddisfano i criteri assegnati.
SOMMA.PIÙ.SE	Somma le celle di un intervallo (è possibile che le celle di un intervallo siano adiacenti o non adiacenti) che soddisfano più criteri.	=SOMMA.PIÙ.SE(int_somma;intervallo_criteri1;criteri1;intervallo_criteri2;criteri2;...), dove: • int_somma (obbligatorio) indica una o più celle da sommare, inclusi numeri o nomi, intervalli o riferimenti di cella contenenti numeri. Le celle vuote e i valori di testo saranno ignorati; • intervallo_criteri1 (obbligatorio) è il primo intervallo in cui valutare i criteri associati; • criteri1 (obbligatorio) sono criteri in forma di numero, espressione, riferimento di cella o testo che determinano quali celle nell'argomento intervallo_criteri1 saranno sommate. I criteri possono essere espressi per esempio come 10%, ">10%", B4, "Vendite" o "10%"; • intervallo_criteri2; criteri2; ... (facoltativi) sono intervalli aggiuntivi e criteri associati.

CURIOSITÀ

L'ordine degli argomenti della funzione SOMMA.PIÙ.SE è diverso da quello degli argomenti della funzione SOMMA.SE. In particolare, l'argomento int_somma è il primo argomento in SOMMA.PIÙ.SE, mentre in SOMMA.SE è il terzo argomento.

Ogni cella nell'argomento int_somma viene sommata solo se per la cella sono veri tutti i criteri corrispondenti specificati. Ogni argomento intervallo_criteri deve contenere lo stesso numero di righe e colonne dell'argomento int_somma.

ESERCIZIO

4 - Usare le funzioni POTENZA, RADQ e SE per risolvere un'equazione di secondo grado

Crea il file 2-Equazioni secondo grado

Predisponiamo il foglio di lavoro per determinare le soluzioni di un'equazione di secondo grado data nella forma $ax^2 + bx + c = 0$.

Inseriamo:

- in A4 l'equazione $3x^2 + x - 1 = 0$ e nella colonna B i relativi coefficienti;
- in E4 =POTENZA(B6;2)-4*B5*B7 per calcolare il valore del discriminante (Figura 7).

Per calcolare le soluzioni occorre valutare il segno del discriminante Δ (delta); inseriamo le relative formule che, nel caso di soluzioni reali, daranno rispettivamente due risultati.

Inseriamo:

- in C9 =SE(E4<=0;"l'equazione non ha soluzioni reali";(-B6-RADQ(E4))/(2*B5));
- in C11 =SE(E4<=0;"l'equazione non ha soluzioni reali";(-B6+RADQ(E4))/(2*B5)).

E4		=POTENZA(B6;2)-4*B5*B7	
	A	B	C
1	RISOLUZIONE DI UN'EQUAZIONE DI SECONDO GRADO		
2	nella forma $ax^2 + bx + c = 0$		
3			
4	$3x^2 + x - 1 = 0$		$\Delta = b^2 - 4ac$
5	a =	3	
6	b =	1	
7	c =	-1	
8			
9	Soluzioni	$x_1 =$	-0,76759
10		$x_2 =$	0,434259
11			

Figura 7

Il test posto nelle due funzioni, verifica se il **delta** è minore o uguale a zero; se la condizione è vera, allora restituisce il messaggio "l'equazione non ha soluzioni reali", altrimenti vengono calcolate le due soluzioni. Elaboriamo il prospetto come mostrato a pagina precedente (Figura 7). Duplichiamo 2 volte il Foglio 1 e rinominiamo ciascun foglio (Figura 8 e 9). Mantenendo la struttura e le formule impostate, in ciascun foglio, in A4 cambiamo l'equazione e nelle celle della zona B5:B7 inseriamo i relativi coefficienti (Figura 8 e 9), quindi verifichiamo i risultati ottenuti.

Figura 8

	A	B	C	D	E	F	G	H
1	RISOLUZIONE DI UN'EQUAZIONE DI SECONDO GRADO							
2	nella forma $ax^2 + bx + c = 0$							
3								
4	$9x^2 - 6x + 1 = 0$		$\Delta = b^2 - 4ac$	0				
5	a =	9						
6	b =	-6						
7	c =	1						
8								
9	Soluzioni	$x_1 =$	l'equazione non ha soluzioni reali					
10		$x_2 =$	l'equazione non ha soluzioni reali					
11								

Figura 9

	A	B	C	D	E	F	G	H
1	RISOLUZIONE DI UN'EQUAZIONE DI SECONDO GRADO							
2	nella forma $ax^2 + bx + c = 0$							
3								
4	$5x^2 - 6x + 1 = 0$		$\Delta = b^2 - 4ac$	16				
5	a =	5						
6	b =	-6						
7	c =	1						
8								
9	Soluzioni	$x_1 =$	0,2					
10		$x_2 =$	1					
11								

ESERCIZIO

5 - Usare la funzione POTENZA per convertire un numero binario in decimale

■ Crea il file 2-Binario_decimale

Predisponiamo il foglio di lavoro per effettuare la conversione di un numero binario (di 5 cifre) in decimale. Ricordiamo che, per convertire un numero binario in decimale, si moltiplica ogni cifra per 2 elevato alla potenza della corrispondente posizione occupata (partendo da destra) e si sommano poi i risultati.

Vediamo le diverse modalità di calcolo che possiamo applicare.

Nel primo caso utilizziamo la funzione POTENZA e l'operatore matematico + (Figura 10). In B4 inseriamo: $=B2*POTENZA(2;B3)+C2*POTENZA(2;C3)+D2*POTENZA(2;D3)+E2*POTENZA(2;E3)+F2*POTENZA(2;F3)$.

Figura 10

	A	B	C	D	E	F	G	H	I	J	K
1	CONVERSIONE DI UN NUMERO BINARIO (5 cifre) IN DECIMALE										
2	Numero binario	1	0	0	1	1					
3	Potenze di 2	4	3	2	1	0					
4	Numero decimale	19									
5											

Nel secondo caso utilizziamo le funzioni SOMMA e POTENZA (Figura 11). In B4 inseriamo: $=SOMMA(B2*POTENZA(2;B3);C2*POTENZA(2;C3);D2*POTENZA(2;D3);E2*POTENZA(2;E3);F2*POTENZA(2;F3))$.

Figura 11

	A	B	C	D	E	F	G	H	I	J	K	L
1	CONVERSIONE DI UN NUMERO BINARIO (5 cifre) IN DECIMALE											
2	Numero binario	1	0	0	1	1						
3	Potenze di 2	4	3	2	1	0						
4	Numero decimale	19										
5												

Nel terzo caso utilizziamo solo gli operatori matematici ^ e + (Figura 12). In B4 inseriamo: $=B2*2^B3+C2*2^C3+D2*2^D3+E2*2^E3+F2*2^F3$.

Come possiamo notare, ogni modalità di calcolo porta sempre allo stesso risultato.

Figura 12

	A	B	C	D	E	F	G
1	CONVERSIONE DI UN NUMERO BINARIO (5 cifre) IN DECIMALE						
2	Numero binario	1	0	0	1	1	
3	Potenze di 2	4	3	2	1	0	
4	Numero decimale	19					
5							

In B10 (Figura 13), applicando una delle modalità sopra descritte, eseguiamo la conversione di un numero binario (di 6 cifre) in decimale.

Figura 13

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
7	CONVERSIONE DI UN NUMERO BINARIO (6 cifre) IN DECIMALE													
8	Numero binario	1	0	0	1	0	1							
9	Potenze di 2	5	4	3	2	1	0							
10	Numero decimale	37												
11														

Figura 14

	A	B	C	D	E	F	G	H
1	CONVERSIONE DI UN NUMERO BINARIO (5 cifre) IN DECIMALE							
2	Numero binario	1	0	0	1	1		
3	Potenze di 2	4	3	2	1	0		
4	Numero decimale	19						
5								
7	CONVERSIONE DI UN NUMERO BINARIO (6 cifre) IN DECIMALE							
8	Numero binario	1	0	0	1	0	1	
9	Potenze di 2	5	4	3	2	1	0	
10	Numero decimale	37						
11								

Elaboriamo i prospetti (Figura 14).

Per abbreviare l'impostazione di questi calcoli, si può utilizzare la specifica funzione (appartenente alla categoria Ingegneria) BINARIO.DECIMALE seguita dal numero binario da convertire (Figura 15).

Figura 15

	A	B	C	D	E	F
1	CONVERSIONE DI UN NUMERO BINARIO (5 cifre) IN DECIMALE					
2	Numero binario	1	0	0	1	1
3	Potenze di 2	4	3	2	1	0
4	Numero decimale	19				
5						

ESERCIZIO

6 - Funzioni matematiche MATRICE.SOMMA.PRODOTTO, ARROTONDA e SOMMA

■ Apri il file 2-Riparto spese comuni presente nei contenuti digitali del volume (hoepiscuola.it)

Uno studio di dentisti associati ripartisce le spese comuni (affitto e utenze) in base al guadagno annuo e al numero di pazienti di ciascun medico. Inseriamo:

• in B5 $=B4/MATR.SOMMA.PRODOTTO(B8:B10;C8:C10)$ per determinare il coefficiente di riparto (Figura 16);

Figura 16

	A	B	C	D
1	Studi medici associati			
2				
3	Riparto spese comuni			
4	Spese complessive	19800		
5	Coefficiente di riparto	0,000318417		
6				
7	Medico	Guadagno annuo	Pazienti	Quota spese
8	Dott. Marchesini	89500	256	
9	Dott.ssa Barbieri	91700	248	
10	Dott. Ghilardi	79850	207	
11	Totale spese			
12				

- in D8 $=ARROTONDA(\$B\$5*B8*C8;2)$ per determinare (arrotondata al secondo decimale) la quota di spesa di ogni medico. Il riferimento alla cella B5 è inserito come assoluto per poter copiare la funzione fino a riga 10;
- in D11 $=SOMMA(D8:D10)$ per verificare il totale delle spese complessive.

Elaboriamo il prospetto come mostrato (Figura 17).

CURIOSITÀ

Per svolgere l'esercizio di riparto composto diretto è necessario:

- calcolare il relativo coefficiente di riparto;
- moltiplicare il coefficiente di riparto per le grandezze per cui si era ripartito (guadagno annuo x numero pazienti).

Figura 17

	A	B	C	D
1	Studi medici associati			
2				
3	Riparto spese comuni			
4	Spese complessive	€	29.860,00	
5	Coefficiente di riparto		0,000480199	
6				
7	Medico	Guadagno annuo	Pazienti	Quota spese
8	Dott. Marchesini	€	89.500,00	256
9	Dott.ssa Barbieri	€	91.700,00	248
10	Dott. Ghilardi	€	79.850,00	207
11	Totale spese			€
12				29.860,00

ESERCIZIO

7 - Funzioni matematiche SOMMA.PIÙ.SE, SOMMA.SE e SOMMA

Apri il file 2-Fatturato agenti presente nei contenuti digitali del volume (hoepiscuola.it)

Un'azienda esegue il resoconto mensile delle vendite effettuate dai propri agenti di commercio determinando per ognuno il fatturato complessivo. Inoltre, calcola il fatturato complessivo di ogni cliente, gli incassi realmente effettuati per ciascuno di essi e gli importi non ancora pagati.

Inseriamo:

- in G3 =SOMMA.SE(\$C\$3:\$C\$14;F3;\$B\$3:\$B\$14) per determinare il fatturato complessivo di ciascun agente (Figura 18). Per agevolare la copia della formula fino a riga 5, le zone C3:C14 e F3:F14 sono state fissate come assolute e come criterio non è stato inserito il cognome dell'agente, ma il riferimento di cella (F3 relativo);

G3						
=SOMMA.SE(\$C\$3:\$C\$14;F3;\$B\$3:\$B\$14)						
	A	B	C	D	E	F
1	Resocontazione mese di Gennaio					
2	Codice Cliente	Fatturato	Agente	Pagamenti	Agente	Fatturato complessivo
3	CL56/ES	€ 9.350,75	Esposito	Effettuato	Esposito	24056,05
4	CL48/DN	€ 9.846,35	Donardi		Donardi	
5	CL32/BK	€ 6.136,90	Becken	Effettuato	Becken	
6	CL56/ES	€ 5.740,85	Esposito		Totale	
7	CL63/ES	€ 4.980,00	Esposito	Effettuato		
8	CL48/DN	€ 6.964,35	Donardi	Effettuato		
9	CL37/BK	€ 5.135,20	Becken		Situazione pagamenti	
10	CL48/DN	€ 6.564,30	Donardi		Codice Cliente	Fatturato complessivo
11	CL48/DN	€ 6.787,25	Donardi	Effettuato	CL56/ES	
12	CL37/BK	€ 6.445,00	Becken		CL63/ES	
13	CL56/ES	€ 3.984,45	Esposito		CL48/DN	
14	CL37/BK	€ 4.125,40	Becken	Effettuato	CL32/BK	

Figura 18

- in H10 =SOMMA.PIÙ.SE(\$B\$3:\$B\$14;\$A\$3:\$A\$14;F10;\$D\$3:\$D\$14;"Effettuato") per determinare l'importo versato dal cliente CL56/ES. Anche in questa formula, per agevolare la copia fino a riga 14, le zone B3:B14, A3:A14 e D3:D14 sono state fissate come assolute e nel primo criterio è stato inserito il riferimento di cella (F10 relativo);
- in I10 =SOMMA.PIÙ.SE(\$B\$3:\$B\$14;\$A\$3:\$A\$14;F10;\$D\$3:\$D\$14;"") per determinare l'importo da versare da parte del cliente CL56/ES. Copiare la formula fino a riga 14 (Figura 19);

I10									
=SOMMA.PIÙ.SE(\$B\$3:\$B\$14;\$A\$3:\$A\$14;F10;\$D\$3:\$D\$14;"")									
	A	B	C	D	E	F	G	H	I
1	Resocontazione mese di Gennaio				Fatturati agenti				
2	Codice Cliente	Fatturato	Agente	Pagamenti	Agente	Fatturato complessivo			
3	CL56/ES	€ 9.350,75	Esposito	Effettuato	Esposito	24.056,05			
4	CL48/DN	€ 9.846,35	Donardi		Donardi	30.162,25			
5	CL32/BK	€ 6.136,90	Becken	Effettuato	Becken	21.842,50			
6	CL56/ES	€ 5.740,85	Esposito		Totale	€ 76.060,80			
7	CL63/ES	€ 4.980,00	Esposito	Effettuato					
8	CL48/DN	€ 6.964,35	Donardi	Effettuato					
9	CL37/BK	€ 5.135,20	Becken		Situazione pagamenti				
10	CL48/DN	€ 6.564,30	Donardi		Codice Cliente	Fatturato complessivo	Importo versato	Importo da versare	
11	CL48/DN	€ 6.787,25	Donardi	Effettuato	CL56/ES	19.076,05	9.350,75	9.725,30	
12	CL37/BK	€ 6.445,00	Becken		CL63/ES	4.980,00	4.980,00	-	
13	CL56/ES	€ 3.984,45	Esposito		CL48/DN	30.162,25	13.751,60	16.410,65	
14	CL37/BK	€ 4.125,40	Becken	Effettuato	CL32/BK	6.136,90	6.136,90	-	
15	Totale	€ 76.060,80			CL37/BK	15.705,60	4.125,40	11.580,20	
16					Totale	€ 76.060,80	€ 38.344,65	€ 37.716,15	

Figura 19

- in G10 =SOMMA(H10:I10) per determinare il fatturato complessivo del cliente CL56/ES;
- nella riga 15 e in G6 calcoliamo i totali.

Elaboriamo le tabelle come mostrato (Figura 19) e nella zona I10:I14 applichiamo la formattazione condizionale per evidenziare le celle contenenti un valore maggiore di zero.

SOMMA 3D o Consolida dati

FUNZIONE	TIPO DI CALCOLO	SINTASSI
SOMMA 3D	Appartiene al gruppo di formule tridimensionali (3D) perché costituite da riferimenti che considerano una terza dimensione oltre le due già disponibili nel foglio di lavoro (righe e colonne). L'impiego di formule tridimensionali è anche definito consolidamento dei dati. Per utilizzare le formule 3D nella cartella è opportuno predisporre tabelle con la stessa struttura; inoltre, i valori da prendere in considerazione, riportati in più fogli, dovranno trovarsi nella stessa cella o zona di celle.	1 Digitare =SOMMA(poi fare clic sulla scheda del primo foglio, quindi clic nella cella da sommare; 2 tenendo premuto il tasto Shift, fare clic sulla scheda dell'ultimo foglio (...) dell'intervallo di fogli contenenti i valori da sommare e digitare la parentesi) per chiudere la formula così strutturata =SOMMA('Foglio1:Foglio...!H3).

ESERCIZIO

8 - Formule tridimensionali: funzione SOMMA 3D

Apri il file 2-Preventivo spese trimestrali presente nei contenuti digitali del volume (hoepiscuola.it)

Sulla base dei dati emersi nell'esercizio 2-Riparto spese comuni supponiamo che sia stato redatto un preventivo di spesa per l'anno successivo, suddividendo le spese comuni di ciascun medico per trimestri.

In un foglio riassuntivo verranno sommati i singoli importi di spesa per determinare la spesa annua complessiva di ciascun medico.

Gli importi di spesa presenti in ciascun foglio sono stati divisi per 4, quanti sono i trimestri in un anno (Figura 20).

CURIOSITÀ

Come nel bilancio preventivo, un preventivo di spesa fa riferimento all'anno successivo a quello in cui esso viene redatto. Esso contempla l'indicazione delle entrate che si prevede di realizzare (in questo caso, reddito annuo di ogni medico e numero di pazienti) e delle spese che si ipotizza di sostenere (affitto e utenze) nell'anno non ancora iniziato.

D8				
=ARROTONDA(\$B\$5*\$B\$8*\$C\$8/4;2)				
	A	B	C	D
1	Studi medici associati			
2				
3	Riparto spese comuni			
4	Spese complessive	€	29.860,00	
5	Coefficiente di riparto		0,000480199	
6				
7	Medico	Guadagno annuo	Pazienti	Quota spese 1° trimestre
8	Dott. Marchesini	€ 89.500,00	256	2.750,58
9	Dott.ssa Barbieri	€ 91.700,00	248	2.730,12
10	Dott. Ghilardi	€ 79.850,00	207	1.984,30
11	Totale spese			€ 7.465,00
12				
	1° trim	2° trim	3° trim	4° trim
	Riassuntivo			

Figura 20

Nel foglio Riassuntivo, in D8 costruisci la formula tridimensionale nel seguente modo:

- digita =SOMMA(quindi fai clic sulla scheda 1°_trim e poi clic in D8;
- tenendo premuto il tasto Shift, fai clic su 4°_trim (ultima scheda dell'intervallo di fogli contenente i valori da sommare)

- fai clic sul pulsante **Invio** (Calc: **Accetta**) e copia la formula fino a **riga 10** (Figura 22).

Figura 21

Figura 22

ESERCIZIO

178 B.2 FOGLIO ELETTRONICO ADV

Figura 25

Esercizio

10 - Funzione di ricerca e riferimento CERCA.ORIZZ

Apri il file 2-Premi fatturato presente nei contenuti digitali del volume (hoepiscuola.it)

Un'azienda predispone il resoconto mensile dei fatturati effettuati dai propri agenti, stabilendo il premio e il giudizio da assegnare in base al fatturato effettuato.

Viene predisposta la **tabella criteri** con i dati che definiscono, in base al punteggio fatturato totale, il premio e il giudizio assegnati. Nella tabella, i **dati di confronto (categoria)** sono disposti in riga; per eseguire la ricerca, quindi, utilizzeremo la funzione **CERCA.ORIZZ**. Inseriamo:

- in E11 =SOMMA.SE(\$A\$2:\$A\$21;D11;\$B\$2:\$B\$21) per calcolare il fatturato complessivo di ciascun agente (Figura 26);
- in F11 =CERCA.ORIZZ(E11;\$D\$2:\$H\$4;2) per ricercare il giudizio assegnato a ciascun agente in base al fatturato effettuato. Nella funzione applichiamo il **riferimento assoluto** agli indirizzi della tabella matrice per copiare la formula fino a riga 14 e mantenere così inalterati i riferimenti di cella;

E11	A	B	C	D	E	F	G	H
1	Agente	Fatturato		TABELLA CRITERI				
2	Centuri Franco	120		Fatturato totale	€ 500,00	€ 1.000,00	€ 1.500,00	€ 3.000,00
3	Borgheri Luca	159		Giudizio	Standard	Discreto	Buono	Ottimo
4	Corsi Alberto	854		Premio	€ 100,00	€ 200,00	€ 300,00	€ 500,00
5	Turri Luigi	138						
6	Borgheri Luca	123						
7	Centuri Franco	102						
8	Turri Luigi	789						
9	Turri Luigi	159						
10	Borgheri Luca	453		TABELLA RIASSUNTIVA				
11	Corsi Alberto	565		Agente	Fatturato totale	Giudizio	Premio	
12	Centuri Franco	224		Borgheri Luca	€ 1.652,00			
13	Turri Luigi	120		Centuri Franco				
14	Borgheri Luca	104		Corsi Alberto				
15	Corsi Alberto	523		Turri Luigi				
16	Centuri Franco	324		Totale				
17	Corsi Alberto	761						
18	Borgheri Luca	813						
19	Turri Luigi	145						
20	Centuri Franco	175						
21	Corsi Alberto	464						
22	Totale							

Figura 26

G11	A	B	C	D	E	F	G	H
1	Agente	Fatturato		TABELLA CRITERI				
2	Centuri Franco	€ 120,00		Fatturato totale	€ 500,00	€ 1.000,00	€ 1.500,00	€ 3.000,00
3	Borgheri Luca	€ 159,00		Giudizio	Standard	Discreto	Buono	Ottimo
4	Corsi Alberto	€ 854,00		Premio	€ 100,00	€ 200,00	€ 300,00	€ 500,00
5	Turri Luigi	€ 138,00						
6	Borgheri Luca	€ 123,00						
7	Centuri Franco	€ 102,00						
8	Turri Luigi	€ 789,00						
9	Turri Luigi	€ 159,00						
10	Borgheri Luca	€ 453,00		TABELLA RIASSUNTIVA				
11	Corsi Alberto	€ 565,00		Agente	Fatturato totale	Giudizio	Premio	
12	Centuri Franco	€ 224,00		Borgheri Luca	€ 1.652,00	Buono	€ 300,00	
13	Turri Luigi	€ 120,00		Centuri Franco	€ 945,00	Standard	€ 100,00	
14	Borgheri Luca	€ 104,00		Corsi Alberto	€ 3.167,00	Ottimo	€ 500,00	
15	Corsi Alberto	€ 523,00		Turri Luigi	€ 1.351,00	Discreto	€ 200,00	
16	Centuri Franco	€ 324,00		Totale	€ 7.115,00			
17	Corsi Alberto	€ 761,00						
18	Borgheri Luca	€ 813,00						
19	Turri Luigi	€ 145,00						
20	Centuri Franco	€ 175,00						
21	Corsi Alberto	€ 464,00						
22	Totale	€ 7.115,00						
23								

Figura 27


- in G11 =CERCA.ORIZZ(E11;\$D\$2:\$H\$4;3) per ricercare il premio assegnato a ciascun agente in base al fatturato effettuato. Nella funzione applichiamo il **riferimento assoluto** agli indirizzi della tabella matrice per copiare la formula fino a riga 14 e mantenere così inalterati i riferimenti di cella;
- in B22 e in E15 calcoliamo i totali ed elaboriamo le tabelle come mostrato (Figura 27).

CURIOSITÀ

La funzione ha il seguente procedimento logico: leggi il dato scritto in F11; cerca tale valore nella prima riga dell'intervallo D2:H4 e, una volta trovato, restituisci il dato contenuto nella cella della rispettiva colonna della riga 2.

TEST AL VOLO

- Nella categoria Statistiche, quale funzione conta le celle che soddisfano tutti i criteri impostati?
☐ SOMMA.PIÙ.SE ☐ CONTA.PIÙ.SE
☐ SOMMA.SE ☐ CONTA.SE
- Nella sintassi della funzione SE, quale simbolo si dovrà inserire come ultimo argomento se si vuole che al verificarsi della condizione falsa la cella rimanga vuota?
☐ <> ☐ "" ☐ =0 ☐ 0
- Nelle funzioni con test condizionali, quale operatore logico usi per esprimere il significato "diverso"?
☐ =>0 ☐ <> ☐ "" ☐ <>
- Quale dei seguenti operatori matematici può essere utilizzato in sostituzione della funzione POTENZA?
☐ " ☐ > ☐ ^ ☐ '




fisso i CONCETTI CHIAVE

1 Completa la seguente mappa inserendo le voci sotto riportate.

didattica inclusiva

CONTA.PIÙ.SE - riferimento - tridimensionali - SOMMA.PIÙ.SE - ricerca - sintassi - SOMMA 3D - Matematiche - SE - Statistiche - CERCA.VERT - Logiche - MATRICE.SOMMA.PRODOTTO



2 Collega i termini della prima colonna con le definizioni riportate nella seconda.

1 SE	A Si utilizza quando nella tabella i dati di confronto (categoria) sono disposti in colonna.	1 ...
2 CERCA.VERT	B Somma le celle di un intervallo che soddisfano più criteri.	2 ...
3 SOMMA.PIÙ.SE	C Utilizza colori e icone per facilitare l'analisi dei dati.	3 ...
4 Formattazione condizionale	D Verifica una particolare condizione, restituendo un valore per vero e un altro per falso	4 ...